

# Clubbing Cods

## A User's Guide to Kodkodi 1.5.2

Jasmin Christian Blanchette  
Institut für Informatik, Technische Universität München

January 23, 2012

### Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Installing the Tool</b>	<b>2</b>
<b>3</b>	<b>First Steps</b>	<b>3</b>
<b>4</b>	<b>Input Format</b>	<b>4</b>
4.1	Lexical Issues . . . . .	4
4.2	Overall Structure . . . . .	6
4.2.1	Problems . . . . .	6
4.2.2	Problem . . . . .	6
4.2.3	Kodkod Options . . . . .	6
4.2.4	Universe Specification . . . . .	7
4.2.5	Relation Bound Specifications . . . . .	7
4.2.6	Integer Bound Specification . . . . .	7
4.2.7	Solve Directive . . . . .	8
4.3	Register Directives . . . . .	8
4.3.1	Tuple Register Directives . . . . .	8
4.3.2	Expression Register Directives . . . . .	8
4.4	Tuple Language . . . . .	9
4.4.1	Tuples . . . . .	9
4.4.2	Tuple Sets . . . . .	9
4.4.3	Tuple Set Operator Precedences and Associativities . . . . .	10
4.5	Expression Language . . . . .	10
4.5.1	Formulas . . . . .	11
4.5.2	Relational Expressions . . . . .	12
4.5.3	Integer Expressions . . . . .	13
4.5.4	Declarations . . . . .	13
4.5.5	Assignments . . . . .	13
4.5.6	Operator Precedences and Associativities . . . . .	14

4.6	Comments . . . . .	14
5	<b>Case Study: Sorting Using Alloy and Kodkodi</b>	<b>15</b>
6	<b>Known Bugs and Limitations</b>	<b>17</b>

## 1 Introduction

Kodkodi is a front-end for the Java library Kodkod [6], a highly-optimized relational model finder developed by the Software Design Group at MIT. Kodkod is based on SAT solving and supports first-order logic with relations, transitive closure, and partial instances. Kodkod forms the basis of version 4 of the Alloy Analyzer [3]. The Kodkodi front-end is designed to make the Kodkod library available to other programming languages than Java.

This manual explains the concrete syntax supported by Kodkodi. It also explains how to install the tool on your workstation. If you use Kodkodi in conjunction with Nitpick for Isabelle/HOL [1], read the installation instructions in the Nitpick manual [2].

Comments and bug reports concerning Kodkodi or this manual should be directed to `blanchette@in.tum.de`.

## 2 Installing the Tool

To install Kodkodi, download and extract the archive `http://isabelle.in.tum.de/~blanchet/kodkodi-1.5.0.tgz`. The `.jar` files for Kodkodi, the Kodkod library, the portable SAT4J solver, and the ANTLR 3.1.1 runtimes are located in the `jar` subdirectory. Kodkodi requires a Java 1.6 virtual machine, normally called `java`. To run Kodkodi, you must add the `.jar` files in this directory to the Java classpath (e.g., by setting the `CLASSPATH` environment variable) and execute

```
java de.tum.in.isabelle.Kodkodi.Kodkodi
```

To recompile Kodkodi, you need a Java compiler and the ANTLR 3.1.1 parser generator tools.

For better performance, it is recommended to install a C or C++ SAT solver. Kodkodi already includes binary versions of MiniSat, CryptoMiniSat, and Lingeling for various platforms in the `jni` subdirectory. Follow the instructions on Kodkod's home page [4] to install other SAT solvers integrated using the JNI, or install a command-line solver of your choice and specify it using the `External` or `ExternalV2` described in §4.2.3.

### 3 First Steps

Kodkodi takes its input from standard input and writes its output to standard output (on success) or standard error (on failure). Examples are provided in the examples directory. When invoked with wrong command-line arguments, Kodkodi displays the usage text:

```
Usage: java de.tum.in.isabelle.Kodkodi.Kodkodi [options]
options:
  -help                Show usage and exit
  -verbose             Produce more output
  -solve-all          Output all solutions for each problem
  -prove              Output minimal unsatisfiable core
  -max-solutions       Maximum number of solutions to generate (default: ∞)
  -clean-up-inst      Remove trivial parts of instance from output
  -max-msecs <num>   Maximum running time in milliseconds
  -max-threads <num> Maximum number of simultaneous threads
                    (default: <machine-dependent-value>)
  -server             Run as TCP server
  -port <number>     Listen to specified port (default: 9128)
```

Example input files are provided in the examples directory:

```
$ export JAVA_LIBRARY_PATH=jni/x86-linux:$JAVA_LIBRARY_PATH
$ de.tum.in.isabelle.Kodkodi.Kodkodi < examples/pigeonhole.kki
*** PROBLEM 1 ***

--OUTCOME--
UNSATISFIABLE

--STATS--
p cnf 54 68
primary variables: 6
parsing time: 65 ms
translation time: 92 ms
solving time: 0 ms
```

(Replace x86-linux with the appropriate platform.)

The output is to be interpreted as follows:

- The two numbers following p cnf indicate the number of propositional variables and CNF clauses.
- The primary variables are those variables that encode Kodkod relations. (The other variables are auxiliary variables.)
- The time spent on a problem is divided between parsing time (the time Kodkodi took to convert the text file into a Kodkod abstract syntax tree), translation time (the time Kodkod took to convert the abstract syntax tree

into a SAT problem), and solving time (the time the SAT solver took to solve the problem).

To obtain all the solutions instead of only the first solution to a problem or set of problems, pass the `-solve-all` option. To obtain only the first  $n$  solutions, pass `-max-solutions  $n$` .

## 4 Input Format

Kodkodi's input format is modeled after the output format of the `toString()` implementations found in the Kodkod library. The operators that are available in Alloy 4 are given the same precedences as they have there.

The grammar is expressed using a variant of Extended Backus-Naur Form. The actual grammar used by Kodkodi is written using ANTLR and can be found in the file `src/Kodkodi.g`.

### 4.1 Lexical Issues

The grammar is based on the following lexical units, or tokens:

```

    WHITESPACE ::= ( ␣ | \n | \r | \t | \v )+
    INLINE_COMMENT ::= // ~ ( \n )* ( \n | eof )
    BLOCK_COMMENT ::= /* .+ */
    NUM ::= [ + | - ] ( 0 | 1 | ... | 9 )+
    STR_LITERAL ::= " ~ ( " | \n )* "
    ATOM_NAME ::= A NAT
    UNIV_NAME ::= u NAT
    OFF_UNIV_NAME ::= u NAT @ NAT
    TUPLE_NAME ::= ( P | T NAT _ ) NAT
    RELATION_NAME ::= ( s | r | m NAT _ ) NAT
    VARIABLE_NAME ::= ( S | R | M NAT _ ) NAT ' ?
    TUPLE_REG ::= $ ( A | P | T NAT _ ) NAT
    TUPLE_SET_REG ::= $ ( a | p | t NAT _ ) NAT
    FORMULA_REG ::= $ f NAT
    REL_EXPR_REG ::= $ e NAT
    INT_EXPR_REG ::= $ i NAT

```

`NAT` abbreviates `0 | (1 | ... | 9) (0 | ... | 9)*`. Whitespace and comments are ignored, except as token separators. In addition to the tokens listed above, various keywords and operators are recognized as tokens. These are shown in bold.

The table below describes the lexical conventions adopted for naming atoms, tuples, relations, variables, and registers.

Token Name	Syntax	Description
<i>ATOM_NAME</i>	$A_j$	Atom at index $j$ in the universe
<i>UNIV_NAME</i>	$un$	Set of atoms $\{A_0, \dots, A(n-1)\}$
<i>OFF_UNIV_NAME</i>	$un@j$	Set of atoms $\{A_j, \dots, A(j+n-1)\}$
<i>TUPLE_NAME</i>	$P_j$	Pair at index $j$ in the pair space associated with the universe
	$T_{n_j}$	$n$ -tuple at index $j$ in the $n$ -tuple space associated with the universe ( $n \geq 3$ )
<i>RELATION_NAME</i>	$s_j$	Plain set number $j$
	$s_j'$	Primed set number $j$
	$r_j$	Plain binary relation number $j$
	$r_j'$	Primed binary relation number $j$
	$mn_j$	Plain $n$ -ary multirelation number $j$ ( $n \geq 3$ )
	$mn_j'$	Primed $n$ -ary multirelation number $j$ ( $n \geq 3$ )
<i>VARIABLE_NAME</i>	$S_j$	Plain set variable number $j$
	$S_j'$	Primed set variable number $j$
	$R_j$	Plain binary relation variable number $j$
	$R_j'$	Primed binary relation variable number $j$
	$Mn_j$	Plain $n$ -ary multirelation variable number $j$ ( $n \geq 3$ )
	$Mn_j'$	Primed $n$ -ary multirelation variable number $j$ ( $n \geq 3$ )
<i>TUPLE_REG</i>	$\$A_j$	One-tuple register number $j$
	$\$P_j$	Pair register number $j$
	$\$T_{n_j}$	$n$ -tuple register number $j$ ( $n \geq 3$ )
<i>TUPLE_SET_REG</i>	$\$a_j$	One-tuple set register number $j$
	$\$p_j$	Pair set register number $j$
	$\$t_{n_j}$	$n$ -tuple set register number $j$ ( $n \geq 3$ )
<i>FORMULA_REG</i>	$\$f_j$	Formula register number $j$
<i>REL_EXPR_REG</i>	$\$e_j$	Relational expression register number $j$
<i>INT_EXPR_REG</i>	$\$i_j$	Integer expression register number $j$

The primed versions of relations and variables are provided for syntactic convenience. They behave exactly as the unprimed versions, except that they live in a different namespace.

## 4.2 Overall Structure

This section presents the overall structure of Kodkodi input files.

### 4.2.1 Problems

$$\text{problems} ::= \text{problem}^*$$

Kodkodi takes a list of “problems” as input.

### 4.2.2 Problem

$$\text{problem} ::= \text{option}^* \text{univ\_spec} \text{tuple\_reg\_directive}^* \text{bound\_spec}^* \text{int\_bound\_spec}^* \\ \text{expr\_reg\_directive}^* \text{solve\_directive}$$

A problem consists of three main parts: a universe specification, a set of bound specifications, and a Kodkod formula to satisfy supplied in a “solve” directive.

Example:

```
univ: u1
bounds s0: {A0}
solve all [S0 : one s0, S1 : one s0] | S0 = S1;
```

### 4.2.3 Kodkod Options

$$\text{option} ::= \text{solver} : \text{STR\_LITERAL}(\text{, STR\_LITERAL})^* \mid \\ \text{symmetry\_breaking} : \text{NUM} \mid \\ \text{sharing} : \text{NUM} \mid \\ \text{bit\_width} : \text{NUM} \mid \\ \text{skolem\_depth} : \text{NUM} \mid \\ \text{flatten} : (\text{true} \mid \text{false}) \mid \\ \text{delay} : \text{NUM}$$

Kodkod supports various options, documented in the `kodkod.engine.config.Options` class [5]. The following solvers are supported:

```
solver: "CryptoMiniSat"
solver: "DefaultSAT4J"
solver: "LightSAT4J"
solver: "Lingeling"
solver: "MiniSat"
solver: "MiniSatProver"
solver: "ZChaffMincost"
solver: "SAT4J" "instance"
solver: "External", "executable", "cnf_file",
```

```

"arg_1", ..., "arg_n"
solver: "ExternalV2", "executable", "temp_input", "temp_output",
"sat_marker", "var_marker", "unsat_marker", "arg_1", ..., "arg_n"

```

For "External", the optional arguments "arg\_1", ..., "arg\_n" are passed before the input file name. For "ExternalV2", they are passed after.

The delay option specifies a delay (expressed in milliseconds) between solving a problem and exiting, if the `-exit-on-success` command-line option is specified. This can be used to grant additional time to other threads so that they have a chance to finish.

#### 4.2.4 Universe Specification

```

univ_spec ::= univ : UNIV_NAME

```

The universe specification fixes the universe's uninterpreted atoms. Kodkodi requires that the atoms are numbered consecutively from  $A_0$  to  $A_{(n-1)}$ .

Examples:

```

univ: u2
univ: u100

```

#### 4.2.5 Relation Bound Specifications

```

bound_spec ::= bounds RELATION_NAME (, RELATION_NAME) :
(tuple_set | [ tuple_set , tuple_set ])

```

A relational bound specification gives a lower and an upper bound for the given relations. If only one bound is specified, it is taken as both lower and upper bound. The lower bound must be a subset of the upper bound.

Examples:

```

bounds s0: {A0}
bounds r2: [{}, {A0 .. A9} -> {A10 .. A19}]

```

#### 4.2.6 Integer Bound Specification

```

int_bound_spec ::= int_bounds : int_bound_seq (, int_bound_seq)*
int_bound_seq ::= [NUM :] [ tuple_set (, tuple_set)* ]

```

An integer bound specification establishes a correspondence between integers and sets of atoms that represent that integer in relational expressions. The syntax makes it possible to specify the bounds of consecutive integers in sequence.

Example:

```
int_bounds: [{A0}, {A1}], 10: [{A2}, {A3}, {A4}]
```

In the above example, 0 is bounded by {A0}, 1 is bounded by {A1}, 10 is bounded by {A2}, 11 is bounded by {A11}, and 12 is bounded by {A4}.

### 4.2.7 Solve Directive

```
solve_directive ::= solve formula ;
```

The “solve” directive tells Kodkod to try to satisfy the given formula.

Example:

```
solve all [S0 : one s0, S1 : one s0] | ! S0 = S1 => no S0.r0 & S1.r0
```

## 4.3 Register Directives

Registers make it possible to use a complex syntactic construct several times without duplicating it. They also help reduce Kodkod’s memory usage and running time.

### 4.3.1 Tuple Register Directives

```
tuple_reg_directive ::= TUPLE_REG := tuple |  
                          TUPLE_SET_REG := tuple_set
```

A tuple register directive assigns a value to a tuple or tuple set register.

Examples:

```
$P0 := [A0, A0]  
$P1 := [A1, A1]  
$t4_0 := {$P0, $P1} -> {$P0, $P1}
```

### 4.3.2 Expression Register Directives

```
expr_reg_directive ::= FORMULA_REG := formula |  
                          REL_EXPR_REG := rel_expr |  
                          INT_EXPR_REG := int_expr
```

Formulas, relational expressions, and integer expressions can also be assigned to registers using an expression register directive. An alternative is to use the let binder inside an expression.

Examples:



```

$f0 := all [S0 : one s0] | s0 in univ
$e5 := (s0 & s1).r1 + (s0 & s2).r2
$i14 := 2 * #($e5) + 1

```

## 4.4 Tuple Language

Kodkod supports partial solutions in the form of bounds on relations. The bound specifications involve tuples and tuple sets.

### 4.4.1 Tuples

$$\begin{aligned}
 \textit{tuple} ::= & [ \textit{ATOM\_NAME} ( , \textit{ATOM\_NAME} )^* ] | \\
 & \textit{ATOM\_NAME} | \\
 & \textit{TUPLE\_NAME} | \\
 & \textit{TUPLE\_REG}
 \end{aligned}$$

An  $n$ -tuple is normally specified using the syntax  $[A_{j_1}, \dots, A_{j_n}]$ . The brackets are optional when  $n = 1$ . Alternatively, tuples can be specified using an index in the  $n$ -tuple space. For example, given the universe  $u_{10}$ , the name  $P_{27}$  refers to the pair  $[A_2, A_7]$ .

Examples:

```

[A0, A1, A5, A20]
A0
P5
$P14

```

### 4.4.2 Tuple Sets

$$\begin{aligned}
 \textit{tuple\_set} ::= & \textit{tuple\_set} ( + | - ) \textit{tuple\_set} | \\
 & \textit{tuple\_set} \& \textit{tuple\_set} | \\
 & \textit{tuple\_set} \rightarrow \textit{tuple\_set} | \\
 & \textit{tuple\_set} [ \textit{NUM} ] | \\
 & \{ \textit{tuple} ( , \textit{tuple} )^* \} | \\
 & \{ \textit{tuple} \dots \textit{tuple} \} | \\
 & \{ \textit{tuple} \# \textit{tuple} \} | \\
 & \textit{none} | \\
 & \textit{all} | \\
 & \textit{UNIV\_NAME} | \\
 & \textit{OFF\_UNIV\_NAME} | \\
 & \textit{TUPLE\_SET\_REG} | \\
 & ( \textit{tuple\_set} )
 \end{aligned}$$

Tuple sets can be constructed in several ways. The +, -, and & operators denote the union, difference, and intersection of two tuple sets, respectively. The -> operator denotes the Cartesian product of two tuple sets. The [] operator projects the tuple set onto the given dimension. Tuple sets can be specified exhaustively by listing all their tuples. If all the tuples have consecutive indices, the range operator .. can be used. Alternatively, if all the tuples occupy a rectangular, cubic, etc., area in the tuple space, they can be specified by passing the lowest and highest corner of the area to the # operator. Finally, none is a synonym for {}, and all denotes the complete tuple set (whose arity is deduced from the context).

Examples:

```
{A1}
{A1, A2} -> {A3, A4}
{[A1, A2] .. [A3, A4]}
{[A1, A2] # [A3, A4]}
$P14
```

#### 4.4.3 Tuple Set Operator Precedences and Associativities

The operator precedences and associativities are given in the table below. Fully bracketed operators are not listed.

Level	Operator Class	Arity	Associativity
1	+ -	Binary	Left-associative
2	&	Binary	Associative
3	->	Binary	Associative
4	[]	Binary	Left-associative

### 4.5 Expression Language

Kodkod supports three types of expression: Boolean expressions (formulas), relational expressions, and integer expressions.

### 4.5.1 Formulas

```
formula ::= (all | some) decls | formula |  
    let assigns | formula |  
    if formula then formula else formula |  
    formula || formula |  
    formula <=> formula |  
    formula => formula |  
    formula && formula |  
    ! formula |  
    ACYCLIC ( RELATION_NAME ) |  
    FUNCTION ( RELATION_NAME , rel_expr -> (one | lone) rel_expr ) |  
    TOTAL_ORDERING ( RELATION_NAME ,  
        ( UNIV_NAME | OFF_UNIV_NAME | RELATION_NAME ) ,  
        ( ATOM_NAME | RELATION_NAME ) ,  
        ( ATOM_NAME | RELATION_NAME ) ) |  
    rel_expr (in | =) rel_expr |  
    int_expr (= | < | <= | > | >=) int_expr |  
    (no | lone | one | some) rel_expr |  
    false |  
    true |  
    FORMULA_REG |  
    ( formula )
```

A formula, or Boolean expression, specifies a constraint involving relations and integers.

Example:

```
some [S0 : some s0] | if S0 in s1 then !$f1 else $i0 <= $i1
```

## 4.5.2 Relational Expressions

```
rel_expr ::= let assigns | rel_expr |  
             if formula then rel_expr else rel_expr |  
             rel_expr (+ | -) rel_expr |  
             rel_expr ++ rel_expr |  
             rel_expr & rel_expr |  
             rel_expr -> rel_expr |  
             rel_expr \ rel_expr |  
             rel_expr ( rel_expr ( , rel_expr)* ) |  
             rel_expr [ int_expr ( , int_expr)* ] |  
             rel_expr . rel_expr |  
             ( ^ | * | ~ ) rel_expr |  
             { decls | formula } |  
             (Bits | Int) [ int_expr ] |  
             iden |  
             ints |  
             none |  
             univ |  
             ATOM_NAME |  
             UNIV_NAME |  
             OFF_UNIV_NAME |  
             RELATION_NAME |  
             VARIABLE_NAME |  
             REL_EXPR_REG |  
             ( rel_expr )
```

A relational expression denotes a relation (set, binary relation, or multirelation). Nearly all operators are identical to those offered by Kodkod, which in turn are modeled after those provided by Alloy. Notable exceptions are the conditional expression `if ... then ... else ...`; the  $r \setminus s$  operator, which is a shorthand for `if no  $r$  then  $s$  else  $r$` ; and finally  $r(s_1, \dots, s_n)$ , which is equivalent to  $s_n.( \dots(s_1.r) \dots )$ .

Example:

```
if #(s0) > 5 then s0.r0 + s1.r1 else none
```

### 4.5.3 Integer Expressions

$$\begin{aligned} \text{int\_expr} ::= & \text{sum } \text{decls} \mid \text{int\_expr} \mid \\ & \text{let } \text{assigns} \mid \text{int\_expr} \mid \\ & \text{if } \text{formula} \text{ then } \text{int\_expr} \text{ else } \text{int\_expr} \mid \\ & \text{int\_expr} (\ll \mid \gg \mid \gg\gg) \text{int\_expr} \mid \\ & \text{int\_expr} (+ \mid -) \text{int\_expr} \mid \\ & \text{int\_expr} (* \mid / \mid \%) \text{int\_expr} \mid \\ & (\# \mid \text{sum}) (\text{rel\_expr}) \mid \\ & \text{int\_expr} \mid \text{int\_expr} \mid \\ & \text{int\_expr} \wedge \text{int\_expr} \mid \\ & \text{int\_expr} \& \text{int\_expr} \mid \\ & (\sim \mid - \mid \text{abs} \mid \text{sgn}) \text{int\_expr} \mid \\ & \text{NUM} \mid \\ & \text{INT\_EXPR\_REG} \mid \\ & (\text{int\_expr}) \end{aligned}$$

An integer expression denotes an integer.

Example:

$$(\text{sum } [S0 : \text{one } s0] \mid \#(S0) * (\#(S0) + 1) / 2) \% 10$$

### 4.5.4 Declarations

$$\begin{aligned} \text{decls} ::= & [ \text{decl} (, \text{decl})^* ] \\ \text{decl} ::= & \text{VARIABLE\_NAME} : (\text{no} \mid \text{lone} \mid \text{one} \mid \text{some} \mid \text{set}) \text{rel\_expr} \end{aligned}$$

The all, some, and sum quantifiers take a list of variable declarations.

Example:

$$[S0 : \text{set } s0, S1 : \text{one } s1]$$

### 4.5.5 Assignments

$$\begin{aligned} \text{assigns} ::= & [ \text{assign} (, \text{assign})^* ] \\ \text{assign} ::= & \text{FORMULA\_REG} := \text{formula} \mid \\ & \text{REL\_EXPR\_REG} := \text{rel\_expr} \mid \\ & \text{INT\_EXPR\_REG} := \text{int\_expr} \end{aligned}$$

The let binder takes a list of register assignments.

Example:

$$[\$f0 := \text{all } [S0 : \text{one } s0] \mid s0 \text{ in univ}, \$i14 := 2 * \#(\$e5) + 1]$$

## 4.5.6 Operator Precedences and Associativities

The operator precedences and associativities are given in the table below. Fully bracketed operators are not listed.

Level	Operator Class	Arity	Associativity
1	all   some   sum   let   if then else	Binary/Ternary	Right-associative
2		Binary	Associative
3	<=>	Binary	Associative
4	=>	Binary	Right-associative
5	&&	Binary	Associative
6	!	Unary	N/A
7	in = < <= > >=	Binary	N/A
8	no lone one some	Unary	N/A
9	<< >> >>>	Binary	Left-associative
10	+ -	Binary	Left-associative
11	* / %	Binary	Left-associative
12	++	Binary	Associative
13	^ &	Binary	Associative
14	->	Binary	Associative
15	\	Binary	Associative
16	(,)	Binary	Left-associative
17	[,]	Binary	Left-associative
18	.	Binary	Left-associative
19	^ * ~ - abs sgn	Unary	N/A

## 4.6 Comments

Comments may be specified as in C++, that is, either as a one line comment starting with // or as a block starting with /\* and ending with \*/.

Examples:

```

/*
  Copyright 2009 Gnomovision, Inc.
*/

univ: u99999 // Don't panic!

```

## 5 Case Study: Sorting Using Alloy and Kodkodi

Although Kodkodi's syntax is similar to Alloy's, there are a few important conceptual differences. Consider the following Alloy specification of integer sorting:

```
abstract sig IntSeq {
  ints : seq Int
}

pred isSorted [s : IntSeq] {
  all i : s.ints.inds - s.ints.lastIdx | s.ints[i] <= s.ints[i + 1]
}

pred isPermutation [pre, post : IntSeq] {
  all p : Int | #{pre.ints.p} = #{post.ints.p}
}

one sig Pre extends IntSeq {}
one sig Post extends IntSeq {}

fact {
  Pre.ints[0] = 7 && Pre.ints[1] = 2 &&
  Pre.ints[2] = 4 && Pre.ints[3] = 3 &&
  Pre.ints[4] = 3 && Pre.ints[5] = 8 &&
  Pre.ints[6] = 5 && Pre.ints[7] = 20 &&
  Pre.ints[8] = 18 && Pre.ints[9] = 1 &&
  Pre.ints[10] = 10 && Pre.ints[11] = 5 &&
  Pre.ints[12] = 7 && Pre.ints[13] = 12 &&
  Pre.ints[14] = 2 && Pre.ints[15] = 19 &&
  Pre.ints[16] = 15 && Pre.ints[17] = 13 &&
  Pre.ints[18] = 11 && Pre.ints[19] = 4
}

run { Pre.isPerm[Post] && Post.isSorted } for 20 seq, 6 int
```

There are two main approaches to representing this in Kodkod:

1. *We could tell the Alloy Analyzer to generate Kodkod-based Java code, call `toString()` on the abstract syntax tree, and fiddle a little bit with the output to make it comply with Kodkodi's input syntax.*

From an Alloy specification, we can generate Java code by choosing "Output Kodkod to file" as the SAT Solver in the Alloy Analyzer's "Options" menu.

Unfortunately, for the example above, the generated code is too large for the Java compiler, which simply bails out. In general, we would need to rename the atoms and relations so that they follow Kodkodi's strict naming conventions and change a few syntactic items.

2. *We ignore the Alloy model and start from scratch in Kodkodi.*

This gives a specification like the following:

```
  solver: "MiniSat"
```

```

bit_width: 6
univ: u21
bounds r0 /* Pre.ints */:
{[A0, A7], [A1, A2], [A2, A4], [A3, A3], [A4, A3], [A5, A8], [A6,
A5], [A7, A20], [A8, A18], [A9, A1], [A10, A10], [A11, A5], [A12,
A7], [A13, A12], [A14, A2], [A15, A19], [A16, A15], [A17, A13],
[A18, A11], [A19, A4]}
bounds r1 /* Post.ints */: [{}, u20->u21]
int_bounds: [{A0}, {A1}, {A2}, {A3}, {A4}, {A5}, {A6}, {A7}, {A8},
{A9}, {A10}, {A11}, {A12}, {A13}, {A14}, {A15}, {A16}, {A17},
{A18}, {A19}, {A20}]
solve FUNCTION(r1, u20->one u21)
&& (all [S0 : one univ] | #(r1.S0) = #(r0.S0))
&& (all [S0 : one u19] | sum(S0.r1) <= sum(Int[sum(S0) + 1].r1));

```

The first two lines,

```

solver: "MiniSat"
bit_width: 6

```

are configuration options. Then we specify that the universe should consist of exactly 21 atoms:

```

univ: u21

```

The atoms are called A0 to A20. Next, we specify the values for the `Pre.ints` relation as a Kodkod bound:

```

bounds r0 /* Pre.ints */:
{[A0, A7], [A1, A2], [A2, A4], [A3, A3], [A4, A3], [A5, A8], [A6,
A5], [A7, A20], [A8, A18], [A9, A1], [A10, A10], [A11, A5], [A12,
A7], [A13, A12], [A14, A2], [A15, A19], [A16, A15], [A17, A13],
[A18, A11], [A19, A4]}

```

In Kodkodi, all binary relations must be called  $r_j$ , where  $j$  is a natural number. The comment is there to remind us that `r0` corresponds to `Pre.ints` in the Alloy specification.

```

bounds r1 /* Post.ints */: [{}, u20->u21]

```

For `Post.ints`, we specify the empty set `{}` as the lower bound and the Cartesian product `{A0 .. A19}->{A0 .. A20}` as the upper bound.

```

int_bounds: [{A0}, {A1}, {A2}, {A3}, {A4}, {A5}, {A6}, {A7}, {A8},
{A9}, {A10}, {A11}, {A12}, {A13}, {A14}, {A15}, {A16}, {A17},
{A18}, {A19}, {A20}]

```

Since we need the integers for addition, we must associate atoms with the integers we need. Here we simply let A0 represent 0, A1 represent 1, and so on.

```

solve FUNCTION(r1, u20->one u21)
&& (all [S0 : one univ] | #(r1.S0) = #(r0.S0))
&& (all [S0 : one u19] | sum(S0.r1) <= sum(Int[sum(S0) + 1].r1));

```



Finally, we specify the formula to solve. The first line ensures that `r1` (i.e., `Post.ints`) is a function rather than an arbitrary relation. The second and third lines are adapted directly from the Alloy specification.

Sorting [7, 2, 4, 3, 3, 8, 5, 20, 18, 1, 10, 5, 7, 12, 2, 19, 15, 13, 11, 4] should give [1, 2, 2, 3, 3, 4, 4, 5, 5, 7, 7, 8, 10, 11, 12, 13, 15, 18, 19, 20]. Let us verify that this is the case by running Kodkodi:

```
*** PROBLEM 1 ***
--OUTCOME--
SATISFIABLE

--INSTANCE--
relations: {r0=[[A0, A7], [A1, A2], [A2, A4], [A3, A3], [A4, A3],
[A5, A8], [A6, A5], [A7, A20], [A8, A18], [A9, A1], [A10, A10],
[A11, A5], [A12, A7], [A13, A12], [A14, A2], [A15, A19], [A16,
A15], [A17, A13], [A18, A11], [A19, A4]], r1=[[A0, A1], [A1, A2],
[A2, A2], [A3, A3], [A4, A3], [A5, A4], [A6, A4], [A7, A5], [A8,
A5], [A9, A7], [A10, A7], [A11, A8], [A12, A10], [A13, A11], [A14,
A12], [A15, A13], [A16, A15], [A17, A18], [A18, A19], [A19, A20]]}
--STATS--
p cnf 8166 29484
primary variables: 420
parsing time: 72 ms
translation time: 359 ms
solving time: 434 ms
```

The result is correct.

## 6 Known Bugs and Limitations

Here are the known bugs and limitations in Kodkodi at the time of writing:

- The `-server` command-line option, which makes Kodkodi run as a TCP server, is limited to a single connection. Furthermore, any error occurring when processing one problem breaks the connection.

## References

- [1] J. C. Blanchette. Nitpick: Yet another counterexample generator for Isabelle/HOL. <http://isabelle.in.tum.de/~blanchet/nitpick.html>.
- [2] J. C. Blanchette. *Picking Nits: A User's Guide to Nitpick for Isabelle/HOL 2010*. Dept. of Informatics, T.U. München, 2010. <http://isabelle.in.tum.de/~blanchet/nitpick-1.2.2/Nitpick/manual/nitpick.pdf>.

- [3] D. Jackson. *Software Abstractions: Logic, Language, and Analysis*. MIT Press, 2006.
- [4] E. Torlak. Kodkod: A constraint solver for relational logic. <http://alloy.mit.edu/kodkod/>.
- [5] E. Torlak. Kodkod API: Class Options. <http://alloy.mit.edu/kodkod/docs/kodkod/engine/config/Options.html>.
- [6] E. Torlak and D. Jackson. Kodkod: A relational model finder. In O. Grumberg and M. Huth, editors, *TACAS 2007*, volume 4424 of *LNCS*, pages 632–647. Springer, 2007.